
ABSTRACT
We describe the use of a flexible meta-interpreter for per-
forming access control checks on deductive databases. The
meta-program is implemented in Prolog and takes as input
a database and an access policy specification. We then pro-
ceed to specialise the meta-program for a given access policy
and intensional database by using the logen partial evalua-
tion system. In addition to describing the programs involved
in our approach, we give a number of performance measures
for our implementation of an access control checker, and we
discuss the implications of using this approach for access
control on deductive databases. In particular, we show that
by using our approach we get flexible access control with
virtually zero overhead.

H.2.7 [Database Management]: Database Administra-
tion—security, integrity, and protection; F.3.2 [Logics and
Meanings of Programs]: Semantics of Programming Lan-
guages—partial evaluation; D.1.2 [Programming Tech-
niques]: Automatic Programming; I.2.2 [Artificial Intel-
ligence]: Automatic Programming; D.1.6 [Programming
Techniques]: Logic Programming; F.4.1 [Mathematical
Logic and Formal Languages]: Mathematical Logic—
logic programming

1. INTRODUCTION
The issue of controlling a user’s ability to exercise access

privileges (e.g., read, write, execute privileges) on a system’s
resources has long been an important issue in Computer Sci-
ence. In recent years, there has been considerable interest
in access control in research and in practice. All aspects
of security have been given particular prominence with the
advent of the Web. In a number of surveys, security issues
have been reported by enterprises as being of paramount
concern when deciding policies on the publication of Web
data, and the availability of Web resources (see, for exam-
ple, [6]). Security issues, including access control issues, will
be of particular importance in the emerging Semantic Web,
and for e-commerce applications (see, for example, [11]).

In recent years, a number of researchers have developed
some sophisticated access control models in which access
control requirements may be expressed by using rules that
are employed to reason about authorised forms of access (see,
for example, [12], [5], and [3]). In these approaches, ac-
cess to resources are expressed by using rules that define
the conditions that must be satisfied in order for a permis-
sion/denial/authorisation to hold. Expressing access control
policies by using rules is natural, and enables many implicit
permissions/denials/authorisations to be expressed in a suc-
cinct manner. However, an important practical issue that
arises with the rule-based approach to access control is the
problem of efficiently evaluating access requests when ac-
cess control requirements are implicitly specified. The prob-
lem of efficiently evaluating access requests with respect to
rule-based specifications of access policies has become espe-
cially important recently, as organizations require increas-
ingly complex forms of access control policies for protecting
resources. These complex forms of policy specifications are
potentially expensive to compute.

For each of the approaches described in [12], [5], and [3],
proposals are made for attempting to ensure that access re-
quests are evaluated efficiently when access control require-
ments are specified implicitly. In [12] and [5], view material-
isation approaches are described for attempting to optimize
access control checks. The motivation for the view mate-
rialisation approach is to make explicit the access control
information that is implicitly defined in rule form. Making
explicit the implicitly specified access control information,
means that access requests can be evaluated by considering
explicitly recorded facts rather than these facts having to
be derived at query evaluation time. Unfortunately, view
materialisation is not so efficient to use when large num-



bers of parametric derivation rules [4] are used to express
access control requirements and when the specification of
access control requirements changes dynamically e.g., when
user session information [3] is used in the course of deciding
whether an access request is authorised.

Rather than using view materialisation techniques, the
approach described in [3] enables access requests to be effi-
ciently evaluated by utilizing constraint logic programming
techniques [19]. The approach described in [3] makes use of
specialised constraint solvers, rather than view materialisa-
tion techniques, for the efficient evaluation of access requests
in situations where large numbers of parametric derivation
rules (e.g., rules that express temporal constraints on user
access) would be expensive to compute, and when changes to
an access policy are performed dynamically as a consequence
of a user’s session management. Nevertheless, the potential
optimization of access requests by using program specialisa-
tion techniques is not considered in [3]. Furthermore, each
of the approaches described in [12], [5], and [3] assumes that
access control is expressed with respect to coarse-grained
data objects (e.g., files and directories), and that an answer
to an access request on a data item is simply whether access
is allowed or not. In contrast, the work in [1] has the signif-
icant computational attraction of exploiting request modifi-
cation techniques to combine the decision on allowing access
with the actual generation of authorised data that may be
released to answer a user’s access request. However, the
approach described in [1] does not exploit specific access
request optimization methods.

In contrast to [12], [5], and [3], we describe an approach to
the problem of access request evaluation where large num-
bers of parametric derivation rules are required in order to
specify access policy requirements; where fine-grained access
to data items is required (e.g., access to atomic formulae);
where the answer to a user access request generates the set
of logical consequences that the user is permitted to see;
and where access control information is to be exploited for
performance gains.

In overview, we describe an access control checker that is
implemented by using a meta-program that is written as a
logic program. The meta-program takes as input an access
control program and a database. The approach that we de-
scribe enables the access control program to be specialised,
in order to reduce the amount of run-time information that
needs to be considered when deciding whether an access re-
quest is authorised. In effect, the approach ensures that a
minimal amount of information is considered at access re-
quest evaluation time. Specifically, the user session infor-
mation that applies at the time of an access request is used
with a form of access control program that is specialised by
using the relatively static information explicitly specified in
the access control program.

In practice, the rules defining an access control policy are
not subject to frequent changes. As such, this relatively
static information may be exploited for program specialisa-
tion. Moreover, an access control request is a request that
is made by a specific (authenticated) user to perform a spe-
cific operation (i.e., read, write, execute, etc.) on a specific
database item. Exploiting the information about a user’s
identity and the access privilege the user wishes to exer-
cise on a database object can be exploited to specialise a
program for access control and hence can be exploited for
computational advantage.

Although meta-interpreters have previously been devel-
oped for efficient constraint checking on databases [17], to
the best of our knowledge, no approach has yet been pro-
posed in the literature for generating specialised access re-
quests via a meta-interpreter that manipulates access re-
quests, access control policies and databases as object level
expressions, and that precompiles access checking for certain
access requests. In this paper, we describe a technique to ob-
tain a specialised access control checker that is more efficient
to use than using a database and access control program di-
rectly because some of the propagation, simplification and
evaluation process is precompiled.

We consider the use of role-based access control (RBAC)
policies [3] for specifying authorised forms of access to data-
base objects. In RBAC, the most fundamental notion is
that of a role. A role is defined in terms of a job func-
tion in an organization (e.g., a doctor role in a medical en-
vironment), and users and access privileges on objects are
assigned to roles. Moreover, access privileges on objects
(i.e., permissions) are assigned to roles (e.g., a doctor has
the permission to change a patient’s prescriptions). RBAC
policies have a number of well documented attractions [23],
and are widely used in practice [9]. Although we restrict
our attention to RBAC policies in this paper, it should be
noted that RBAC is a more general form of access control
model than the discretionary access control and mandatory
access control approaches that predate RBAC [8], and the
approach that we describe can be used with more powerful
access control methods than RBAC (e.g., the status-based
access control model [2]). It follows that our approach is
widely applicable.

We represent an RBAC policy by using a logic program.
The use of logic programs for representing access control
policies has been recognised in a number of recent works
(see, for example, [12] and [3]). Logic programs enable access
policies to be expressed by using high-level declarative lan-
guages for which formally well defined semantics and oper-
ational methods with attractive theoretical properties (e.g.,
termination) are known to exist.

The rest of this paper is organized as follows. In Section 2,
some basic notions in logic programming and the logen
partial evaluation system (which has been improved since
the system used in [17]) are briefly described. In Section 3,
we briefly describe an RBAC model, and the formulation
of RBAC policies by using logic programs. In Section 4 we
describe the metaprogram that we use for the evaluation of
access requests on databases with respect to a formulation of
an RBAC policy. In Section 5, we present some performance
measures for an implementation of our approach, and we
discuss the results that we report. Finally, in Section 6, some
conclusions are drawn and suggestions for further work are
made.

2. PRELIMINARIES
In this section, we briefly describe some syntactic and se-

mantic issues relating to logic programming. We then give a
brief overview of partial evaluation using the logen system.

2.1 Syntax and Semantics
The RBAC model and the RBAC policies that we de-

scribe in later sections are expressed in the language of
(function-free) normal clause form logic (Datalog¬), with
certain predicates in the alphabet Σ of the language having



a fixed intended interpretation. As we only admit function-
free clauses, the only terms of relevance to Σ will be con-
stants and variables. Hereafter, we denote variables that
appear in clauses by using symbols that appear in upper
case (at least the first character), and constants will be de-
noted by lower case symbols.

A normal clause is a formula of the form:

C ← A1, . . . , Am,¬B1, . . . ,¬Bn (m ≥ 0, n ≥ 0).

The head, C, of the clause above is a single atom. The body
of the clause (i.e., A1, . . . , Am, ¬B1, . . . ,¬Bn) is a conjunc-
tion of literals. Each Ai literal (i ∈ {1, .., m}) is a positive
literal; each ¬Bj literal (j ∈ {1, .., n}) is a negative literal.
In the case of a negative literal, the relevant type of negation
is negation as failure [7]. A clause with an empty body is an
assertion or a fact. A clause with a non-empty head and a
non-empty body is a rule. A relational database is a set of
facts; a normal deductive database is a set of normal clauses.
The set of facts in a deductive database ∆ is referred to as
the extensional part of ∆, (the EDB of ∆), and the set of
rules in ∆ is referred to as the intensional part of ∆ (the
IDB of ∆).

In our representation of a database, a fact of the form
p(c1, . . . , cn) (where each subscripted p is an arbitrary n-
place predicate and ci,∀i ∈ {1, . . . , n}, are constants) is
represented as an atom of the following form:

fact(p(c1, . . . , cn)).

A clause of the following form (where each subscripted p
is an arbitrary n-place predicate and each subscripted t is a
term)

p1(t1, . . . , tn)← p2(ti, . . . , tj), . . . , pm(tk, . . . , tl).

is represented in our databases by using an atom of the
following form:

rule(p1(t1, . . . , tn), [p2(ti, . . . , tj), . . . , pm(tk, . . . , tl)]).

The access control programs that we consider are always
locally stratified (a realistic assumption for most practical
policies) and hence have a unique perfect model [20]. Hav-
ing a 2-valued model theoretic semantics is important for
ensuring that authorised forms of access are unambiguously
specified.

2.2 Partial Evaluation and thelogen System
Partial evaluation [14] is a source-to-source program trans-

formation technique that specialises programs by fixing part
of the input of some source program P and then pre-com-
puting those parts of P that only depend on the known part
of the input. The so-obtained transformed programs are less
general than the original, but can be much more efficient.
The part of the input that is fixed is referred to as the static
input, while the remainder of the input is called the dynamic
input.

Partial evaluation is especially useful when applied to in-
terpreters. In that setting, the static input is typically the
object program being interpreted, while the actual call to
the object program is dynamic. Partial evaluation can then
produce a more efficient, specialised version of the inter-
preter, which is sometimes akin to a compiled version of the
object program [10].

The logen system [16] is a so-called offline partial evalua-
tor for Prolog, i.e., specialisation is divided into two phases,
as depicted in Figure 1:

• First a binding-time analysis (BTA for short) is per-
formed which, given a program and an approximation
of the input available for specialisation, approximates
all values within the program and generates annota-
tions that steer (or control) the specialisation process.

• A (simplified) specialisation phase, which is guided by
the result of the BTA.

Specialised
Program

Source
Program BTA

Annotated
Source
Program

Partial
Evaluator

Specialised
Program

Static
Input

Dynamic
Input Output

Figure 1: Offline Partial Evaluation

Because of the preliminary BTA, the specialisation pro-
cess itself can be performed very efficiently, with predictable
results. Also, as shown in [15], the logen system is well
suited to specialise interpreters, something that we will aim
to exploit in our approach.

3. RBAC POLICIES AS LOGIC PROGRAMS
In this section, we describe a simple type of RBAC policy

that may be used to protect the information in databases.
More specifically, the one type of policy that we describe
here is based on the RBACP

H2A model that is formally de-
fined in [3]. We only consider one type of access control pol-
icy in this paper because our principal concern is to describe
the generalities of using a meta-programming approach for
access request checking, access policy program specialisation
by logen, and performance evaluation. It should be noted,
however, that any of the policies from [3] may be represented
by using our meta-programming approach to access control
checking, with minor modifications.

We call an access control program that is defined in terms
of the RBACP

H2A model, an RBACP
H2A program. This type

of program is a finite set of normal clauses specified with
respect to a domain of discourse that includes:

• A set U of users.

• A set O of objects.



• A set A of access privileges.

• A set R of roles.

In an RBACP
H2A program, a user is specified as being

assigned to a role by using definitions of a 2-place ura pred-
icate, and the assignment of an access privilege on an ob-
ject to a role is expressed by using definitions of a 3-place
pra predicate in the RBACP

H2A program. The semantics of
these predicates in an arbitrary RBACP

H2A program Π may
be expressed thus:

• Π |= ura(u, r) iff user u ∈ U is assigned to role r ∈ R;

• Π |= pra(a, o, r) iff the access privilege a ∈ A on object
o ∈ O is assigned to the role r ∈ R.

By separating the assignment of users to roles from the
assignment of permissions to roles it is possible for user-
role and permission-role assignments to be changed inde-
pendently of each other in implementations of RBACP

H2A

policies. Thus, access policy maintenance is simplified (rel-
ative to the discretionary access control policies that were,
until recently, used as a matter of course to help to protect
the information in databases).

In the RBACP
H2A model, specified in [3], an RBACP

H2A

role hierarchy is defined as a (partially) ordered set of roles.
The ordering relation is a role seniority relation. In an
RBACP

H2A program Π, a 2-place predicate senior to(ri, rj)
is used to define the seniority ordering between pairs of roles.
That is, the role ri ∈ R is a more senior role (or more pow-
erful role) than role rj ∈ R. If ri is senior to rj then any
user assigned to the role ri has at least the permissions that
users assigned to rule rj have. More formally, the semantics
of the senior to relation may be expressed thus:

• Π |= senior to(ri, rj) iff the role ri ∈ R is senior to
the role rj ∈ R in an RBACP

H2A role hierarchy.

The senior to relation may be defined as the reflexive-
transitive closure of an irreflexive-intransitive binary rela-
tion ds. The semantics of ds may be expressed, in terms of
an RBACP

H2A program Π, thus:

• Π |= ds(ri, rj) iff the role ri ∈ R is senior to the role
rj ∈ R in an RBACP

H2A role hierarchy defined in Π
and ¬∃rk ∈ R [ds(rk, rj) ∧ ds(ri, rk)] where rk 6= ri

and rk 6= rj .

Assuming the lattice of role hierarchies to be complete,
an RBACP

H2A role hierarchy is defined by the following set
of clauses (in which ‘ ’ is an anonymous variable):

senior to(R1, R1) ← ds(R1, ).
senior to(R1, R1) ← ds( , R1).
senior to(R1, R2) ← ds(R1, R2).
senior to(R1, R2) ← ds(R1, R3), senior to(R3, R2).

In RBAC models generally, senior roles are assumed to
inherit the access privileges on objects that are assigned to
junior roles in an RBACP

H2A role hierarchy. An RBACP
H2A

role hierarchy enables many authorisations to be implicitly
defined, thus simplifying the expression of access control
policies.

In RBAC, users activate and deactivate roles in the course
of session management [3] as required to perform the tasks
associated with a job function. In [3], the notion of a user

ui ∈ U activating a role rj ∈ R in a session is represented
by using a set of rules of the following form:

active(U, R) ← activate(U, R), C.

In this context, a user ui requests to be active in a role rj by
appending an activate(ui, rj) fact to an RBACP

H2A program
via a GUI. An active(ui, rj) fact will be implicitly appended
to an RBACP

H2A program whenever ui has requested to be
active in a role rj and the set of conditions C, on ui’s acti-
vation of the role rj is satisfied. Any activate assertion that
enables the user ui to be active in role rj may be retracted
by ui when ui no longer wishes to be active in rj , and all of
the activate assertions for a user are automatically retracted
when the user logs off of the system.

An authorisations clause [3] is used to define that a user
ui ∈ U has the ak ∈ A access privilege on object ol ∈ O. In
the case of RBACP

H2A programs, the authorisations clause
is defined thus:

permitted(U, A, O) ← ura(U, R1),
active(U, R1),
senior to(R1, R2),
pra(A, O, R2).

The rule that defines permitted is used to express that a
user U may exercise the A access privilege on object O if: U
is assigned to the role R1, U is active in R1, R1 is senior to
a role R2 in an RBACP

H2A role hierarchy, and R2 has been
assigned the A access privilege on O.

In the context of specialising an RBACP
H2A program Π,

we note that the definitions of ura, pra, senior to and ds
are part of the object level information that is used to pro-
tect the object level database in our approach. Moreover,
the sets of clauses defining the extensions of the ura, pra,
ds and senior to relations are static relative to the set of
active atoms that are implicit in Π. That is, the set of
active facts will change dynamically as users activate and
deactivate roles. The aim of our approach is to specialise
RBACP

H2A programs to enable efficient access control checks
to be performed by only considering user session informa-
tion expressed via the set of active facts that is current at
the time of a user’s access control request.

4. THE META-INTERPRETER
In this section, we describe the meta-interpreter that we

propose for efficient access request evaluation on deductive
databases that are protected by RBACP

H2A programs. We
restrict our attention to a consideration of read access.

The following Prolog code is part of the meta-interpreter
that is used to execute the RBACP

H2A programs that we
have described for access control:

holds_read(User,not(Object)) :-
\+(holds_read(User,Object)).

holds_read(_User,Object) :- built_in(Object).

holds_read(User,Object) :-
permitted(User,read,Object),
fact(Object), call(Object).

holds_read(User,Object) :-
permitted(User,read,Object),
rule(Object,Body),
l_holds_read(User,Body).

l_holds_read(_U,[]).



l_holds_read(U,[H|T]) :- holds_read(U,H),
l_holds_read(U,T).

built_in(’=’(X,X)).
built_in(’is’(X,Y)) :- X is Y.

holds(U,O) :- holds_read(U,O).

The full code can be found in Appendix B. The definition
of permitted that is assumed in this example is that we
described above for implementing an RBACP

H2A policy, to
wit:

permitted(User,Op,Obj) :- ura(User,Role),
active(User,Role),
senior_to(Role,R2),
pra(R2,Op,Obj).

This paper only considers the definition of authorisations
by permitted/3, as its goal is to apply RBACP

H2A policies.
Any number of alternative definitions of permitted may be
used to implement different access policies (see [3] for other
definitions of authorisation clauses that may be used), and
do not require any modifications to our meta-interpreter in
order to process those access requests.

Example 1. Consider an RBACP
H2A program Π with the

following sets of facts:

DS = {ds(r1, r2)}.

ACTIV E = {active(u1, r1), active(u2, r2)}.

URA = {ura(u1, r1), ura(u1, r2), ura(u2, r2)}.

PRA = {pra(r1, read, s( )), pra(r2, read, p( )),
pra(r2, read, q( , )), pra(r1, read, r( , ))}.

Moreover, suppose that Π is used to protect the following
database ∆ in which p and s are EDB predicates and p and
q are IDB predicates:

fact(p(X)).
fact(s(X)).
rule(q(X, Y ), [p(X), p(Y )]).
rule(r(X, Y ), [q(X, Y ), s(X)]).

The access request holds_read(u1,q(A,B)) by user u1 to
read all instances of q from ∆, can be specialised by logen
into:

holds_read(u1,q(A,B)) :- holds_read__0(A,B).

permitted__1(B,C) :- active(u1,r1).
permitted__1(D,E) :- active(u1,r2).

permitted__4(B) :- active(u1,r1).
permitted__4(C) :- active(u1,r2).

holds_read__3(B) :- permitted__4(B), p(B).

holds_read__0(B,C) :- permitted__1(B,C),
holds_read__3(B),
holds_read__3(C).

By inspection, it is possible to see that the effect of such a
specialisation is to reduce a predicate like permitted, which
is defined in terms of the relatively static predicates ura,
pra, ds and senior to, to tests on the run-time information
that is generated in the course of session management, i.e.,
active facts.

5. PERFORMANCE MEASURES
In this section, we give some performance measures for

the meta-programming approach that we propose for eval-
uating access requests on deductive databases that are pro-
tected by using an RBACP

H2A program. Our testing in-
volved comparing the evaluation of access requests on (i)
a non-specialised, and (ii) a logen specialised RBACP

H2A

meta-programs. For comparison’s sake, we also measured
versions of the RBACP

H2A that have no access control. These
versions are implemented directly as Prolog clauses and hence
needed no meta-interpreter to run.

The RBACP
H2A programs that we use in our tests have

included a definition of the senior to relation that represents
an RBACP

H2A role hierarchy with 53 roles arranged as a
complete lattice, and with each node/role of outdegree 3 or
indegree 3. The senior to relation has been materialised
into a set of 312 pairs of ground binary assertions1.

We have experimented with variants of the RBACP
H2A

role hierarchy by increasing the depth of the role lattice.
The summation that follows describes the number of pairs
of roles in the senior to relation as defined by the RBACP

H2A

role hierarchy that we use in testing:

N + 2

d−1/2X
i=1

3i + (Nd ∗ P>1)

In the summation above, N is the total number of nodes
in the role lattice, d is the depth of the lattice, Nd is the
number of nodes at depth d in the lattice, and P>1 is the
number of paths of length 2 or greater from a node at depth
d.

The unique bottom element in all of the RBACP
H2A role

hierarchies that we use in testing is assigned the read per-
mission on all of the logical consequences of the databases
that we use in testing. Moreover, our testing is based on a
single user that is assigned to the most senior role/unique
top element in the RBACP

H2A role hierarchies/complete lat-
tices that are used in our testing. Access requests are eval-
uated for this user. Our choice of user-role assignment and
permission-role assignments imply that our tests are based
on a worst-case scenario that involves the maximum amount
of inheritance of permissions whenever an access request is
evaluated.

The queries that we use in testing involve computing two
binary relations tcp and cycle, and a unary relation q. The
tcp relation is the transitive closure of a 2-place predicate p;
the cycle relation involves computing a transitive closure in
order to determine elements in the reflexive closure of p; the
definition of q is a variant of the well-known win program.2

The tcp program was chosen for inclusion in testing because
of its practical significance; cycle was chosen because it in-
volves some expensive recursive processing; the q program
was chosen because it combines recursion and negation, and
is a useful benchmark test for performance studies.

The definitions of the tcp, cycle and q predicates are ex-
pressed in our database thus:

1In this case, we use a partial materialisation approach such
that only the role hierarchy is materialised (but not the au-
thorisations).
2The win program describes a two-player game in which a
player wins if his or her opponent has no move to make. The
formalisation of this two-person game may be expressed by
the clause: win(X)← move(X, Y ),¬win(Y ).



tcp(X, Y ) ← p(X, Y ).
tcp(X, Y ) ← p(X, Z), tcp(Z, Y ).

cycle(X, Y ) ← p(X, Y ).
cycle(X, Y ) ← cycle(X, Z), p(Z, Y ).

q(X) ← p(X, Y ),¬q(Y ).

The 2-place p predicate is defined by a set of 2495 facts.
A total of 499 p facts are used to represent the chain:

p(a1, a2), p(a2, a3), . . . , p(a498, a499), p(a499, a500).

An additional 1996 p facts are used to achieve a fan-out
factor of 5 [21]. That is, for each p fact with the first argu-
ment ai, where 1 6 i 6 499, there are four p facts with the
second argument of p equal to the value bj , where 1 6 j 6 4.
For example, p(a1, b1), p(a1, b2), p(a1, b3), p(a1, b4). For the
cycle program, at the nth call to cycle, a chain of (n− 1) el-
ements in the transitive closure of p is computed, and hence
the goal clause p(an, an−1) is evaluated. An additional fact
p(a500, a1) is added to the 2495 p facts used with tcp to
represent the end of the cycle.

The successful tcp(a1, a500) query that we use in our test-
ing involves computing a 500 element chain starting from
the element a1 and ending with the element a500. To eval-
uate the tcp query by using SLD-resolution, a search space
comprising 499 SLD-trees with root ← tcp(an, a500), where
1 6 n 6 499, was generated. Each of these 499 SLD-trees
spawns 5 subtrees; 4 of which fail, and one that succeeds.
The four failing cases have a bj value (1 6 j 6 4) as the
second argument of a p fact; the succeeding subtree ter-
minates with an answer clause of the form p(as, at) where
t = s + 1, 1 6 s 6 499 and 2 6 t 6 500. The cycle(a1, a1)
query used involves computing every chain from a1 to aw

(2 6 w 6 500) in the transitive closure of p, until p(a500, a1)
succeeds and hence p(a1, a1) succeeds. The failing query in
our suite of tests (tcp(a1, a501)) is an attempt to compute a
501 element chain that terminates at the element a501. The
successful q(a1) query involves generating 499 failing SLD-
trees for the 499 evaluations of the ¬q(cm) subgoal, where
1 6 m 6 499. The one successful SLD-derivation is gener-
ated from the ground clause: q(a1)← r(a1, c500),¬q(c500).

The results of the testing of our example queries are sum-
marised in Table 1 (for the non-specialised case), and Ta-
bles 2 and 3 (for the specialised case). The queries denoted
by Q1, Q2, Q3 and Q4 in these tables have the following
meanings:

• Q1 is the successful tcp(a1, a500) query run 10 times;

• Q2 is the failed tcp(a1, a501) query run 5 times;

• Q3 is the successful cycle(X, Y ) query (all 145,850 so-
lutions);

• Q4 is the successful q(X) query (all 1,170 solutions).

The query times are expressed in seconds, and are usually
averaged over several runs. The time needed to generate the
compiler from the interpreter (i.e., performing the second

Futamura projection [10]) was 0.040s. The prior binding-
time analysis was performed (once and for all) by hand us-
ing logen’s new graphical interface that allows easy annota-
tion and provides colouring feedback on static and dynamic
parts.3 To achieve the good results it was essential to fol-
low the approach from [15] (see also Appendix C). Timings
were obtained on a Powerbook G4 1Ghz, 1GB SDRAM, with
SICStus Prolog 3.11.0 and Mac OS X 10.3.2. Runtimes for
XSB were obtained on the same machine using XSB Prolog
2.6. In our experiments, we make use of XSB’s distinctive
feature: it terminates for both recursive and non recursive
datalog programs. This mechanism is known as tabling in
XSB Prolog [22], and has been proved very useful in deduc-
tive databases. Tabling allows, for instance, the evaluation
of query Q3, which only XSB Prolog can run ensuring ter-
mination.

Query With Without Overhead
RBACP

H2A RBACP
H2A

Q1 (SICStus) 0.135 s 0.003 s 0.132 s
Q1 (XSB) 0.100 s 0.000 s 0.100 s
Q2 (SICStus) 1.372 s 0.004 s 1.368 s
Q2 (XSB) 0.100 s 0.000 s 0.100 s
Q3 (XSB) 1.460 s 1.080 s 0.380 s
Q4 (SICStus) 9.64 s 0.060 s 9.580 s
Q4 (XSB) 0.109 s 0.010 s 0.099 s

Table 1: Average retrieval times for the non-
specialised case.

Table 1 shows how much overhead is introduced by the ac-
cess control policy. For example, query Q3 that takes 1.08
seconds to retrieve information takes an extra 0.38 seconds
when access control is performed. Ideally, we want to min-
imise the overhead introduced by the RBACP

H2A policy. By
specialisation of the meta-interpreter, we achieve a speedup
that considerably reduces this overhead, as illustrated in Ta-
ble 2. It can be observed that after applying the logen tool,
the average retrieval time is improved by a factor of up to
42. In all cases the retrieval time after specialisation falls
between the average times of the two previous approaches,
i.e., with and without access control.

Query Specialisation Average Speedup
Time Runtime

Q1 0.010 s 0.007 s 19.3
Q2 0.010 s 0.032 s 42.88
Q4 0.010 s 0.950 s 10.15

Q′
1 0.010 s 0.003 s 45

Q′
2 0.010 s 0.004 s 343

Q′
4 0.010 s 0.060 s 120.5

Table 2: Retrieval times (running in SICStus) for
the specialised case.

3An automatic binding-time analysis is in the final stages
of implementation (as can be guessed from the screenshot
in Appendix D) and it will hopefully be able to annotate
our interpreter automatically. However, it is acceptable to
perform the BTA by hand, as the annotation only has to be
generated once and it is independent of the database as well
as the access control policy.



There is, of course, a penalty introduced by this approach:
the specialisation time, i.e., the time it takes logen to fig-
ure out a specialised version of the meta-interpreter with an
RBACP

H2A policy. However, Table 2 shows that adding to-
gether the average runtime and the specialisation time does
not exceed the original times. By adjusting the annotations
(i.e., marking more calls as unfoldable), a more aggressive
specialisation can be obtained. This is shown in the second
part of the table, where Q′

i is the same query as Qi, but
taking the senior to clause into account as well (which is
likely to remain unchanged for a long time).

Query Specialisation Average Speedup
Time Runtime

Q1 0.010 s 0.026 s 3.85
Q2 0.010 s 0.024 s 4.17
Q3 0.010 s 1.220 s 1.20
Q4 0.010 s 0.016 s 6.81

Q′
1 0.010 s 0.010 s 10

Q′
2 0.010 s 0.008 s 12.5

Q′
3 0.010 s 1.130 s 1.29

Q′
4 0.010 s 0.013 s 8.38

Table 3: Retrieval times (running in XSB) for the
specialised case.

Table 3 shows how the previous results compares when
the Prolog Engine includes tabling. It can be observed that
for the more aggressive criteria the time is reduced and even
reaches, in most cases, the ideal without-access-control fig-
ure aimed (i.e., overhead=0). For query Q3 the specialised
interpreter (see Appendix A) is actually almost identical to
the database without access control.4 We have thus actu-
ally achieve what is called “Jones optimality” [13, 14, 18,
15] (called the “optimality criterion” in [14]). The only
drawback of the aggressive specialisation is that each time
senior to changes there is an overhead of 10ms for speciali-
sation (as well as the time needed to load the new specialised
interpreter, which was around 10ms in our experiments).
Since this clause does not change very often, we are not
paying a high price in terms of access control flexibility.

Our testing is based on the scenario where a user inherits
all access privileges on all objects (i.e., logical consequences)
from the most junior role in the RBACP

H2A role hierarchy. In
practice, access request evaluation will involve significantly
less permission inheritance, and user access to data objects
will be far more constrained than we have considered in our
testing. We consider only this scenario because access con-
trol requirements will be highly application specific. The
more specific access control restrictions that will apply in
practice will enable logen to specialise access control pro-
grams to a much greater extent than we have considered,
and therefore more impressive speedup can be expected in
practical applications.

6. CONCLUSIONS AND FURTHER WORK
We have described a partial-deduction approach for spe-

cialising access control checking on deductive databases. Our
approach uses the logen system to specialise RBACP

H2A

4We believe the fact that our specialised interpreter runs
slightly slower is probably due to caching issues.

programs. The results of our experiments using the logen
system, have revealed that program specialisation produces
significant improvements in access request evaluation times
on deductive databases protected by RBACP

H2A programs.
In fact, by using access control information for specialisa-
tion, it is possible to evaluate access requests on deductive
databases as efficiently as evaluating the same requests with-
out processing access control information (in case the user is
actually allowed to access the information). Hence, our ap-
proach makes it possible to incorporate access control checks
into access request evaluations on deductive databases with-
out incurring any overheads.

There are a number of additional issues to investigate in
the context of optimizing access requests on policy infor-
mation in P2P and B2B applications. It would also be in-
teresting to apply our approach to emerging access control
models for controlling access to Web resources (see, for ex-
ample, [2]). We intend to investigate these issues in future
work.

APPENDIX

A. SPECIALISED INTERPRETER FOR Q3

A series of tests are undertaken in Section 5, showing the
efficiency of RBACP

H2A programs after specialising our in-
terpreter using logen. In this section we provide the ac-
tual code obtained from the specialiser, for both aggressive
and non-aggressive specialisation. Annotating the senior to
clause of permitted/3 as a rescall, i.e., adding it to the
residual code so that it is not evaluated in the specialisation
process, leads to the non-aggressive approach shown below:

bench__0 :-

ensure_loaded(database_cycle),
abolish_all_tables, cputime(A),
b2__1,
cputime(B), C is B-A, print(C), nl.

b2__1 :-
seniorto(r1, r53),
holds_read_rule__2(_, _),
fail.

b2__1.
:- table holds_read_rule__2/2.
holds_read_rule__2(A, B) :-

seniorto(r1, r53),
p(A, B).

holds_read_rule__2(A, B) :-
seniorto(r1, r53),
holds_read_rule__2(A, C),
seniorto(r1, r53),
p(C, B).

By considering the senior to clause as unfold, i.e., being
computed in the specialisation, the resulted program may
be more efficient, at the expense of having to re-specialise
each time a parameter to this clause changes. This slightly
more aggressive result is shown as follows:

bench__0 :-
ensure_loaded(database_cycle),
abolish_all_tables, cputime(A),
b2__1,
cputime(B),
C is B-A, print(C), nl.

b2__1 :-
holds_read_rule__2(_, _),
fail.

b2__1.
:- table holds_read_rule__2/2.
holds_read_rule__2(A, B) :-

p(A, B).
holds_read_rule__2(A, B) :-

holds_read_rule__2(A, C),
p(C, B).

Observe that holds read rule 2 is isomorphic to the cycle
predicate, hence Jones-optimality [13, 14, 18, 15] has been
achieved.

B. THE FULL INTERPRETER
Below is the full code of the interpreter, including a predi-

cate bench used for benchmarking our query Q3. The predi-
cates for queries Q1, Q2, and Q4 are very similar. The code
below is intended for XSB Prolog, minor modifications were
done for SICStus (e.g., replacing cputime by statistics).

ura(steve,r1).

active(steve,r1).

pra(r53,read,p(_,_)).
pra(r53,read,cycle(_,_)).
pra(r53,read,tcp(_,_)).
pra(r53,read,q(_)).

:- table holds_read/2.

holds_read(User,not(Object)) :-
\+(holds_read(User,Object)).

holds_read(_User,Object) :- built_in(Object).
holds_read(User,Object) :-

permitted(User,read,Object), fact(Object),
call(Object).



Figure 2: Snapshot of a logen session

holds_read(User,Object) :-
permitted(User,read,Object), derived(Object),
holds_read_rule(User,Object).

holds_read_rule(User,Object) :- rule(Object,Body),
l_holds_read(User,Body).

l_holds_read(_U,[]).
l_holds_read(U,[H|T]) :-

holds_read(U,H),
l_holds_read(U,T).

built_in(’=’(X,X)).
built_in(’is’(X,Y)) :- X is Y.

holds(U,O):- holds_read(U,O).

permitted(User,Op,Obj) :-
ura(User,Role), active(User,Role),
seniorto(Role,R2), pra(R2,Op,Obj).

fact(p(_X,_Y)).
derived(cycle(_,_)).
derived(tcp(_,_)).
derived(q(_)).

rule(cycle(X1,X2),[p(X1,X2)]).
rule(cycle(X1,X2),[cycle(X1,X3),p(X3,X2)]).

rule(tcp(X1,X2),[p(X1,X2)]).
rule(tcp(X1,X2),[p(X1,X3),tcp(X3,X2)]).
rule(q(X),[p(X,Y),not(q(Y))]).

% For benchmarking query Q3:
b2 :- holds_read(steve,cycle(_,_)),fail.
b2.
bench :- ensure_loaded(’database_cycle’),

abolish_all_tables, cputime(T1),
b2,
cputime(T2), R is T2-T1, print(R),nl.

C. THE ANNOTATED INTERPRETER
Bellow is the annotation file as used in the experiments

(aggressive settings). Our GUI interface (see Appendix D)
enables users to view and edit this file in a friendly fashion.

Note that the use of the nonvar annotation was essential
to obtain good specialisation results (see also [15]). Also
observe that a custom type mycall was added so as to avoid
throwing away information within a negated call.

logen(ura, ura(steve,r1)).
logen(active, active(steve,r1)).
logen(pra, pra(r53,read,p(_,_))).
logen(pra, pra(r53,read,cycle(_,_))).



logen(pra, pra(r53,read,tcp(_,_))).
logen(pra, pra(r53,read,q(_))).
(:-true).
logen(holds_read, holds_read(A,not(B))) :-

resnot(logen(memo,holds_read(A,B))).
logen(holds_read, holds_read(_,A)) :-

logen(unfold, built_in(A)).
logen(holds_read, holds_read(A,B)) :-

logen(unfold, permitted(A,read,B)),
logen(unfold, fact(B)),
logen(rescall, call(B)).

logen(holds_read, holds_read(A,B)) :-
logen(unfold, permitted(A,read,B)),
logen(unfold, derived(B)),
logen(unfold, holds_read_rule(A,B)).

logen(holds_read_rule, holds_read_rule(A,B)) :-
logen(unfold, rule(B,C)),
logen(unfold, l_holds_read(A,C)).

logen(l_holds_read, l_holds_read(_,[])).
logen(l_holds_read, l_holds_read(A,[B|C])) :-

logen(memo, holds_read(A,B)),
logen(unfold, l_holds_read(A,C)).

logen(built_in, built_in(A=A)).
logen(built_in, built_in(A is B)) :-

logen(unfold, A is B).
logen(holds, holds(A,B)) :-

logen(unfold, holds_read(A,B)).
logen(permitted, permitted(A,B,C)) :-

logen(unfold, ura(A,D)),
logen(unfold, active(A,D)),
logen(unfold, seniorto(D,E)),
logen(unfold, pra(E,B,C)).

logen(fact, fact(p(_,_))).
logen(derived, derived(cycle(_,_))).
logen(derived, derived(tcp(_,_))).
logen(derived, derived(q(_))).
logen(rule, rule(cycle(A,B),[p(A,B)])).
logen(rule, rule(cycle(A,B),[cycle(A,C),p(C,B)])).
logen(rule, rule(tcp(A,B),[p(A,B)])).
logen(rule, rule(tcp(A,B),[p(A,C),tcp(C,B)])).
logen(rule, rule(q(A),[p(A,B),not(q(B))])).

logen(b2, b2) :-
logen(memo, holds_read(steve,cycle(_,_))),
logen(rescall, fail).

logen(b2, b2).
logen(bench, bench) :-

logen(rescall, ensure_loaded(database_cycle)),
logen(rescall, abolish_all_tables),
logen(rescall, cputime(A)),
logen(unfold, b2),
logen(rescall, cputime(B)),
logen(rescall, C is B-A),
logen(rescall, print(C)),
logen(rescall, nl).

:- type mycall = (not(nonvar) ; nonvar).
:- filter ura(dynamic, dynamic).
:- filter active(dynamic, dynamic).
:- filter pra(dynamic, dynamic, dynamic).
:- filter holds_read_rule(static, type(mycall)).
:- filter holds_read(static, type(mycall)).
:- filter built_in(dynamic).
:- filter holds(static, type(mycall)).
:- filter permitted(dynamic, dynamic, dynamic).
:- filter fact(dynamic).
:- filter rule(dynamic, dynamic).

D. THE LOGEN TOOL IN ACTION
The purpose of this section is more of an illustrative na-

ture rather than actually presenting any theoretical nor em-
pirical result. In this section we show the actual phase of
specialising the RBACP

H2A program by means of the lo-
gen system. As shown in Figure 2, logen is built with
a Graphical User Interface (GUI) which facilitates the spe-
cialisation of logic programs. This snapshot illustrates how
RBACP

H2A program were annotated and specialised in the
context of this framework. On the one hand, the source code
of the database meta-interpreter (left window) is annotated
by unrolling a list of options for each clause in a predicate.
On the other, filter declarations are typed in the far right
window, in order to guide the specialisation process. This
allows logen to, after just pressing a button, specialise the
program for which the RBACP

H2A policy has been optimised
towards a particular query (i.e., the program is specialised
according to the IDB of ∆).


